
Change 1 

	Function name
	f_EUTRA_RRC_Procedure_Latency ()

	Reason for change
	In the execution of function f_EUTRA_RRC_Procedure_Latency(), it is sometimes observed that the calculated time for the uplink grant has already elapsed before the grant is configured to the SS, mostly due to delays in the execution of function f_SS_ConfigUL_HARQ_IndMode().

	Summary of change
	Rearrange the function so that f_EUTRA_ULGrantConfiguration_Common() is called for UL grant configuration before f_SS_ConfigUL_HARQ_IndMode()

	TTCN module
	EUTRA_CommonProcedures.ttcn

	MCC160 Comment
	Accepted as improvement


Before change
	  function f_EUTRA_RRC_Procedure_Latency(EUTRA_CellId_Type p_CellId,

                                          template (value)   SRB_COMMON_REQ p_SRB_COMMON_REQ_1,

                                          template (omit)    SRB_COMMON_REQ p_SRB_COMMON_REQ_2 := omit,

                                          template (present) SRB_COMMON_IND p_SRB_COMMON_IND_1,

                                          template           SRB_COMMON_IND p_SRB_COMMON_IND_2 := omit,

                                          boolean p_IsFDD,

                                          integer p_N,

                                          charstring p_TestcaseStep,

                                          charstring p_UE_Cat,

                                          //@sic R5-153973 to  be set to true for UE capability sic@

                                          boolean p_BigULGrant := false) runs on EUTRA_PTC return SRB_COMMON_IND

  { //@sic R5-152105 sic@ @sic R5s150734 R5-153973 R5-167569: p_UE_Cat->charstring sic@

    var integer v_DL_RetransmissionCnt := 0;

    var SYSTEM_IND v_SYSTEM_IND;

    var SRB_COMMON_IND v_SRB_COMMON_IND;

    var boolean v_T1Flag := false;

    var boolean v_T3Flag := false;

    var boolean v_T4Flag := (not isvalue(p_SRB_COMMON_REQ_2));     // we only check for a second UL PDU when there is a second DL PDU

    var integer v_NoOfGrants := 5;

    var integer v_GrantCycle;

    var integer v_DelayGrantToData;

    var SubFrameTiming_Type v_T0; // Sending of DL RRC PDU(s)

    var SubFrameTiming_Type v_T1; // HARQ ACK for DL RRC PDU(s)

    var SubFrameTiming_Type v_T2; // First UL grant

    var SubFrameTiming_Type v_T3; // First UL RRC PDU received

    var SubFrameTiming_Type v_T4; // Second UL RRC PDU received

    var SubFrameTiming_Type v_T5; // Last UL grant

    var SubFrameTiming_Type v_Tnow;

    var TimingInfo_Type v_TimingInfo;

    var integer v_UlGrantOffset; //@sic R5-144678 R5s141213 sic@

    var integer v_I_MCS:=0;      // Modulation and Coding Scheme Index @sic R5s150734 sic@

    var integer v_N_PRB:=1;      // PRBS @sic R5s150734 sic@

   …………………………
    v_T2 := f_EUTRA_TimingInfoAdd(v_T0, p_N + v_UlGrantOffset);            /* T2: first UL grant is assigned; sub-sequent UL grants are to cope with DL retransmissions

                                                                               @sic R5-144678 R5s141213 sic@  */

    f_SS_ConfigUL_HARQ_IndMode(p_CellId, tsc_L1Mac_IndicationMode_Enable, v_TimingInfo);       /* enable indication of HARQ ack/nack

                                                                                                    @sic R5s120024: scheduled configuration sic@ */

    //@sic R5s120769 sic@

    SRB.send(p_SRB_COMMON_REQ_1);                                                              /* send (first) DL PDU at T0 */

    if (isvalue(p_SRB_COMMON_REQ_2)) {

      SRB.send(p_SRB_COMMON_REQ_2);                                                            /* send second DL PDU at T0 */

    }

    v_T5 := f_EUTRA_TimingInfoAdd(v_T0, v_GrantCycle * (v_NoOfGrants - 1));                 /* T5: last UL grant is sent out; @sic R5s130951 sic@ */

    v_Tnow := f_EUTRA_GetCurrentTiming(p_CellId);

    //@sic R5s170303 int2char -> int2str sic@

    select (p_UE_Cat) {         // Table 7.1.7.1.1.3.2-1: Maximum TBsize  for different UE categories @sic R5s150734 sic@

      //@sic R5-151413 sic@

      case (int2str(0), "m1") { // MAX TB size 1000 64QAM Not supported @sic R5s170220 change 1sic@

        v_I_MCS := 14 ;          // Itbs = 13

        v_N_PRB := 4;            // results in TB size of 1000

      }

      case (int2str(1)) {                // MAX TB size 5160 64QAM Not supported

        v_I_MCS := 15;          // Itbs = 14

        v_N_PRB := 18;          // results in TB size of 5160

      }

      case else {

        if (p_BigULGrant) { //@sic R5-153973 sic@

          v_I_MCS := 20; //ITbs=19, 16 QAM modulation

          v_N_PRB := f_ConvertUL_BandwidthToInteger(f_EUTRA_CellInfo_GetUL_ChBandwidth(p_CellId));

          if (str2int(p_UE_Cat) == 2 and v_N_PRB > 50) { // the MAX UL TBsize supported by Cat 2 is 25456

            v_N_PRB := 58; //truncate N_PRB allocated to result in TBsize of 25456

          }

        } else { //Normal grant

          v_I_MCS := 20;

          v_N_PRB := 25;

        }

      }

    }

    f_EUTRA_ULGrantConfiguration_Common(p_CellId,

                                        cs_TimingInfo_EUTRA_NB(v_T2),

                                        cs_DciInfo_CcchDcchDtchUL_Explicit(v_I_MCS, v_N_PRB,-, f_EUTRA_FirstRB_UL (p_CellId)), //@sic R5s120099r1  R5s130903 R5s140497 R5s150734 R5s170438 sic@

                                        cs_UplinkTimeAlignment_Synch_None,

                                        cs_UL_GrantConfig_NoOfGrants(v_GrantCycle, v_NoOfGrants));
    t_Timer.start (int2float(f_SubFrameTiming_Duration(v_Tnow, v_T5) + 50) / 1000.0);     /* 50ms added to be at the safe side */

    alt {

      [] SYSIND.receive (car_UL_HARQ_IND(p_CellId, ?, nack)) {                                                 /* HARQ NACK */

        v_DL_RetransmissionCnt := v_DL_RetransmissionCnt + 1;

        repeat;

      }

      [] SYSIND.receive (car_HarqError_IND(p_CellId)) -> value v_SYSTEM_IND {                                  /* HARQ error indication */

        if (ischosen(v_SYSTEM_IND.Indication.HarqError.UL)) {

          f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in latency test");

        } else {

          repeat;

        }

      }

      [not v_T1Flag] SYSIND.receive (car_UL_HARQ_IND(p_CellId, ?, ack)) -> value v_SYSTEM_IND {                /* HARQ ACK => T1 */

        v_T1 := v_SYSTEM_IND.Common.TimingInfo.SubFrame;

        f_SS_ConfigUL_HARQ_IndMode(p_CellId, tsc_L1Mac_IndicationMode_Disable, -, tsc_NoCnfReq);               /* disable indication of HARQ ack/nack; @sic R5s120201: no confirmation requested sic@ */

        v_T1Flag := true;

        repeat;

      }

      [not v_T3Flag] SRB.receive(p_SRB_COMMON_IND_1) -> value v_SRB_COMMON_IND {                               /* First UL RRC PDU => T3 */

        v_T3 := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame;

        v_T3Flag := true;

        repeat;      // => wait for timeout to be sure that there are no UL grants pending anymore

      }

      [not v_T4Flag] SRB.receive(p_SRB_COMMON_IND_2) -> value v_SRB_COMMON_IND {                               /* Second UL RRC PDU => T4 */

        if (not v_T3Flag) {

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "Incorrect order of UL PDUs");

        }

        v_T4 := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame;

        v_T4Flag := true;

        repeat;      // => wait for timeout to be sure that there are no UL grants pending anymore

      }

      [] t_Timer.timeout {                                                                                     /* timeout => after T5 */

        // exit the alt

      }

    }

    if (not v_T1Flag) {

      f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "no HARQ ACK for DL PDU");

    }

    if (not v_T3Flag) {

      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "no UL PDU");

    }

    if (not v_T4Flag) {

      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "no second UL PDU");

    }

    // Latency check:

    if ((f_SubFrameTiming_Duration(v_T1, v_T3) > p_N + v_DelayGrantToData) or                                     /* check first RRC procedure regarding latency requirenments */

        (isvalue(p_SRB_COMMON_REQ_2) and (f_SubFrameTiming_Duration(v_T1, v_T4) > p_N + v_DelayGrantToData))) {   /* check second RRC procedure regarding latency requirenments */

      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep);

    } else {

      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, p_TestcaseStep);

    }

    return v_SRB_COMMON_IND;

  }


After Change 
	  /*

   * @desc      send DL RRC PDU(s), receive corresponding UL RRC PDU(s) and check latency requirement given by p_N;

   *            there is one RRC procedure (8.2.1.5) or two (8.2.1.6); in the latter case there are 2 ASP to be sent (having the same timing info) and 2 ASP to be received

   *            HARQ retransmisions are compensated for the DL but will cause inconc for UL.

   * @param     p_CellId

   * @param     p_SRB_COMMON_REQ_1  .. ASP to be sent containing valid timing information and DL RRC PDU

   * @param     p_SRB_COMMON_REQ_2  (default value: omit; only present if there is a second parallel RRC procedure as in 8.2.1.6)

   * @param     p_SRB_COMMON_IND_1  .. ASP with expected UL RRC PDU and optionally NAS PDU

   * @param     p_SRB_COMMON_IND_2  (default value: omit; only present if there is a second parallel RRC procedure as in 8.2.1.6)

   * @param     p_IsFDD

   * @param     p_N                 .. N acc. to e.g. 36.523-1 cl. 8.2.1.6.2; N = 10, 15, 20 i.e. multiple of 5

   * @param     p_TestcaseStep

   * @param     p_UE_Cat            .. highest (!) UE category

   * @param     p_BigULGrant        (default value: false)

   * @return    SRB_COMMON_IND      .. latest IND-ASP

   * @status    APPROVED (LTE, LTE_A_R10_R11)

   */

  function f_EUTRA_RRC_Procedure_Latency(EUTRA_CellId_Type p_CellId,

                                          template (value)   SRB_COMMON_REQ p_SRB_COMMON_REQ_1,

                                          template (omit)    SRB_COMMON_REQ p_SRB_COMMON_REQ_2 := omit,

                                          template (present) SRB_COMMON_IND p_SRB_COMMON_IND_1,

                                          template           SRB_COMMON_IND p_SRB_COMMON_IND_2 := omit,

                                          boolean p_IsFDD,

                                          integer p_N,

                                          charstring p_TestcaseStep,

                                          charstring p_UE_Cat,

                                          //@sic R5-153973 to  be set to true for UE capability sic@

                                          boolean p_BigULGrant := false) runs on EUTRA_PTC return SRB_COMMON_IND

  { //@sic R5-152105 sic@ @sic R5s150734 R5-153973 R5-167569: p_UE_Cat->charstring sic@

    var integer v_DL_RetransmissionCnt := 0;

    var SYSTEM_IND v_SYSTEM_IND;

    var SRB_COMMON_IND v_SRB_COMMON_IND;

    var boolean v_T1Flag := false;

    var boolean v_T3Flag := false;

    var boolean v_T4Flag := (not isvalue(p_SRB_COMMON_REQ_2));     // we only check for a second UL PDU when there is a second DL PDU

    var integer v_NoOfGrants := 5;

    var integer v_GrantCycle;

    var integer v_DelayGrantToData;

    var SubFrameTiming_Type v_T0; // Sending of DL RRC PDU(s)

    var SubFrameTiming_Type v_T1; // HARQ ACK for DL RRC PDU(s)

    var SubFrameTiming_Type v_T2; // First UL grant

    var SubFrameTiming_Type v_T3; // First UL RRC PDU received

    var SubFrameTiming_Type v_T4; // Second UL RRC PDU received

    var SubFrameTiming_Type v_T5; // Last UL grant

    var SubFrameTiming_Type v_Tnow;

    var TimingInfo_Type v_TimingInfo;

    var integer v_UlGrantOffset; //@sic R5-144678 R5s141213 sic@

    var integer v_I_MCS:=0;      // Modulation and Coding Scheme Index @sic R5s150734 sic@

    var integer v_N_PRB:=1;      // PRBS @sic R5s150734 sic@

    timer t_Timer;

    if (p_IsFDD) {

      v_GrantCycle := 8;          // HARQ RTT = 8

      if (pc_FDD_TypeB_HalfDuplex) //@sic R5-182302 R5s180242 sic@

      {

        v_GrantCycle := 10;

      }

      v_DelayGrantToData := 5;    // UL data is always sent 4 + 1 TTIs after the corresponding grant @sic R5-144678 sic@

      v_UlGrantOffset  := 1;       //@sic R5-144678  1 TTI added due toR5-144678 sic@

    } else {

      v_GrantCycle := 10;         // HARQ RTT = 10 for SN = 6

      v_DelayGrantToData := 8;    // UL data is sent 4 + 3 + 1 TTIs after the corresponding grant @sic R5-144678 sic@

      v_UlGrantOffset  := 2;      /* @sic R5-144678 R5s141213 1 TTI added due toR5-144678, but that results in sub frame =5 not suitable for grant in TDD config 1, hence 1 more sub frame added to result in SN 6  sic@ */

    }

    //Get timing reference for calculating expected HARQ ACK i.e. when the performance measurement start

    v_TimingInfo := valueof(p_SRB_COMMON_REQ_1.Common.TimingInfo);

    v_T0 := v_TimingInfo.SubFrame; /* T0: DL PDU is sent the first time; subframe number shall be 4

                                    *     => possible retransmissions are at

                                    *        T0 + 8,  T0 + 16, T0 + 24, T0 + 32    for FDD

                                    *        T0 + 10, T0 + 20, T0 + 30, T0 + 40    for TDD */

    if (pc_ue_CategoryDL_M1) //@sic R5-182302 R5s180242 sic@

    {

      p_SRB_COMMON_REQ_1.Common.TimingInfo.SubFrame.Subframe.Number := 9;

      if (isvalue(p_SRB_COMMON_REQ_2)) { //@sic R5s180409 sic@

        p_SRB_COMMON_REQ_2.Common.TimingInfo.SubFrame.Subframe.Number := 9;

      }

      v_T0.Subframe.Number  := 1;

      v_T0.SFN.Number := (v_T0.SFN.Number + 1) mod 1024;  //@sic R5s180439 sic@

    }

    v_T2 := f_EUTRA_TimingInfoAdd(v_T0, p_N + v_UlGrantOffset);            /* T2: first UL grant is assigned; sub-sequent UL grants are to cope with DL retransmissions

                                                                               @sic R5-144678 R5s141213 sic@  */

    //@sic R5s170303 int2char -> int2str sic@

    select (p_UE_Cat) {         // Table 7.1.7.1.1.3.2-1: Maximum TBsize  for different UE categories @sic R5s150734 sic@

      //@sic R5-151413 sic@

      case (int2str(0), "m1") { // MAX TB size 1000 64QAM Not supported @sic R5s170220 change 1sic@

        v_I_MCS := 14 ;          // Itbs = 13

        v_N_PRB := 4;            // results in TB size of 1000

      }

      case (int2str(1)) {                // MAX TB size 5160 64QAM Not supported

        v_I_MCS := 15;          // Itbs = 14

        v_N_PRB := 18;          // results in TB size of 5160

      }

      case else {

        if (p_BigULGrant) { //@sic R5-153973 sic@

          v_I_MCS := 20; //ITbs=19, 16 QAM modulation

          v_N_PRB := f_ConvertUL_BandwidthToInteger(f_EUTRA_CellInfo_GetUL_ChBandwidth(p_CellId));

          if (str2int(p_UE_Cat) == 2 and v_N_PRB > 50) { // the MAX UL TBsize supported by Cat 2 is 25456

            v_N_PRB := 58; //truncate N_PRB allocated to result in TBsize of 25456

          }

        } else { //Normal grant

          v_I_MCS := 20;

          v_N_PRB := 25;

        }

      }

    }

    f_EUTRA_ULGrantConfiguration_Common(p_CellId,

                                        cs_TimingInfo_EUTRA_NB(v_T2),

                                        cs_DciInfo_CcchDcchDtchUL_Explicit(v_I_MCS, v_N_PRB,-, f_EUTRA_FirstRB_UL (p_CellId)), //@sic R5s120099r1  R5s130903 R5s140497 R5s150734 R5s170438 sic@

                                        cs_UplinkTimeAlignment_Synch_None,

                                        cs_UL_GrantConfig_NoOfGrants(v_GrantCycle, v_NoOfGrants));

    f_SS_ConfigUL_HARQ_IndMode(p_CellId, tsc_L1Mac_IndicationMode_Enable, v_TimingInfo);       /* enable indication of HARQ ack/nack

                                                                                                    @sic R5s120024: scheduled configuration sic@ */

    //@sic R5s120769 sic@

    SRB.send(p_SRB_COMMON_REQ_1);                                                              /* send (first) DL PDU at T0 */

    if (isvalue(p_SRB_COMMON_REQ_2)) {

      SRB.send(p_SRB_COMMON_REQ_2);                                                            /* send second DL PDU at T0 */

    }

    v_T5 := f_EUTRA_TimingInfoAdd(v_T0, v_GrantCycle * (v_NoOfGrants - 1));                 /* T5: last UL grant is sent out; @sic R5s130951 sic@ */

    v_Tnow := f_EUTRA_GetCurrentTiming(p_CellId);
  /* MOVED 

  //@sic R5s170303 int2char -> int2str sic@

    select (p_UE_Cat) {         // Table 7.1.7.1.1.3.2-1: Maximum TBsize  for different UE categories @sic R5s150734 sic@

      //@sic R5-151413 sic@

      case (int2str(0), "m1") { // MAX TB size 1000 64QAM Not supported @sic R5s170220 change 1sic@

        v_I_MCS := 14 ;          // Itbs = 13

        v_N_PRB := 4;            // results in TB size of 1000

      }

      case (int2str(1)) {                // MAX TB size 5160 64QAM Not supported

        v_I_MCS := 15;          // Itbs = 14

        v_N_PRB := 18;          // results in TB size of 5160

      }

      case else {

        if (p_BigULGrant) { //@sic R5-153973 sic@

          v_I_MCS := 20; //ITbs=19, 16 QAM modulation

          v_N_PRB := f_ConvertUL_BandwidthToInteger(f_EUTRA_CellInfo_GetUL_ChBandwidth(p_CellId));

          if (str2int(p_UE_Cat) == 2 and v_N_PRB > 50) { // the MAX UL TBsize supported by Cat 2 is 25456

            v_N_PRB := 58; //truncate N_PRB allocated to result in TBsize of 25456

          }

        } else { //Normal grant

          v_I_MCS := 20;

          v_N_PRB := 25;

        }

      }

    }

    f_EUTRA_ULGrantConfiguration_Common(p_CellId,

                                        cs_TimingInfo_EUTRA_NB(v_T2),

                                        cs_DciInfo_CcchDcchDtchUL_Explicit(v_I_MCS, v_N_PRB,-, f_EUTRA_FirstRB_UL (p_CellId)), //@sic R5s120099r1  R5s130903 R5s140497 R5s150734 R5s170438 sic@

                                        cs_UplinkTimeAlignment_Synch_None,

                                        cs_UL_GrantConfig_NoOfGrants(v_GrantCycle, v_NoOfGrants));
*/
    t_Timer.start (int2float(f_SubFrameTiming_Duration(v_Tnow, v_T5) + 50) / 1000.0);     /* 50ms added to be at the safe side */

    alt {

      [] SYSIND.receive (car_UL_HARQ_IND(p_CellId, ?, nack)) {                                                 /* HARQ NACK */

        v_DL_RetransmissionCnt := v_DL_RetransmissionCnt + 1;

        repeat;

      }

      [] SYSIND.receive (car_HarqError_IND(p_CellId)) -> value v_SYSTEM_IND {                                  /* HARQ error indication */

        if (ischosen(v_SYSTEM_IND.Indication.HarqError.UL)) {

          f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in latency test");

        } else {

          repeat;

        }

      }

      [not v_T1Flag] SYSIND.receive (car_UL_HARQ_IND(p_CellId, ?, ack)) -> value v_SYSTEM_IND {                /* HARQ ACK => T1 */

        v_T1 := v_SYSTEM_IND.Common.TimingInfo.SubFrame;

        f_SS_ConfigUL_HARQ_IndMode(p_CellId, tsc_L1Mac_IndicationMode_Disable, -, tsc_NoCnfReq);               /* disable indication of HARQ ack/nack; @sic R5s120201: no confirmation requested sic@ */

        v_T1Flag := true;

        repeat;

      }

      [not v_T3Flag] SRB.receive(p_SRB_COMMON_IND_1) -> value v_SRB_COMMON_IND {                               /* First UL RRC PDU => T3 */

        v_T3 := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame;

        v_T3Flag := true;

        repeat;      // => wait for timeout to be sure that there are no UL grants pending anymore

      }

      [not v_T4Flag] SRB.receive(p_SRB_COMMON_IND_2) -> value v_SRB_COMMON_IND {                               /* Second UL RRC PDU => T4 */

        if (not v_T3Flag) {

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "Incorrect order of UL PDUs");

        }

        v_T4 := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame;

        v_T4Flag := true;

        repeat;      // => wait for timeout to be sure that there are no UL grants pending anymore

      }

      [] t_Timer.timeout {                                                                                     /* timeout => after T5 */

        // exit the alt

      }

    }

    if (not v_T1Flag) {

      f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "no HARQ ACK for DL PDU");

    }

    if (not v_T3Flag) {

      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "no UL PDU");

    }

    if (not v_T4Flag) {

      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "no second UL PDU");

    }

    // Latency check:

    if ((f_SubFrameTiming_Duration(v_T1, v_T3) > p_N + v_DelayGrantToData) or                                     /* check first RRC procedure regarding latency requirenments */

        (isvalue(p_SRB_COMMON_REQ_2) and (f_SubFrameTiming_Duration(v_T1, v_T4) > p_N + v_DelayGrantToData))) {   /* check second RRC procedure regarding latency requirenments */

      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep);

    } else {

      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, p_TestcaseStep);

    }

    return v_SRB_COMMON_IND;

  }


